



This spreadsheet included many common formulas and tricks we used to develop our CalcSheets. User has access to view all the formulas and see examples of how they can be used in spreadsheets.

## EXCEL FORMULA TRAINING

(Standard Version)

### BASIC

AND & OR

SUM & SUMIF

COUNT & COUNTIF

ROUND

Basic Formulas

TEXT

Shortcut Keys

### INTERMEDIATE

V-LOOKUP

V-LOOKUP (Example)

MATCH

INDEX

IF & NESTED IFs

TEXT & NUMBER in One Cell

DATE

### ADVANCED

Sum Every Other Rows

Multiple Criteria Match

SumProduct

Table & Slicers

Strings In Cell

2D ARRAY SEARCH

Miscellaneous

**COUNT (For cell with numbers or dates)**

**=COUNT(range)**

Use this function to count cells

*Example: Count number of cells that contain numbers. Date is a number. Text will not be counted.*

1
5
3/25/2014
Data
Test
3

Formula Used: =COUNT(C8:C12)

**=COUNT(range1, range2)**

Use for separated cells

2	2
2	2
2	2
2	
7	

Formula Used: =COUNT(C19:C22,E19:E21)

**COUNTA (For Nonblank cells)**

**=COUNTA(range1)**

*Example: Count number of cells that contain numbers or text.*

1
5
3/25/2014
Data
4

Formula Used: =COUNTA(C29:C33)

**COUNTIF (Match)**

**=COUNTIF(range, "Match")**

*Example: Count number of cells with match "jim"*

1
jim
jim
jim
2345
3

Formula Used: =COUNTIF(C39:C43, "Jim")

**COUNTIFS (Multiple Criteria)**

**=COUNTIFS(Range1, Criteria1, Range2, Criteria2,...)**

*Example: Count number of values less than 20 and greater than 10*

8
11
10
11
15
21
3

Formula Used: =COUNTIFS(C50:C55,"<20",C50:C55,">10")

**COUNTIF (For Greater than or equal to 10)**

**=COUNTIF(F52:F56, ">=10")**

*Example: Count number of cells with values greater than or equal to 10*

1
15
12
8
9
2

Formula Used: =COUNTIF(C62:C66, ">=10")

### COUNTIF (Cells with Data)

**=COUNTIF(F52:F56, "<>")**

*Example: Count number of cells with data*

1
15
abc
9
4

Formula Used: =COUNTIF(C74:C78, "<>")

### COUNTBLANK (Cells without Data)

**=COUNTBLANK(Range)**

*Example: Count number of cells that are empty*

1
15
8
2

Formula Used: =COUNTBLANK(C85:C89)

### Count only if Col.1 is text and Col.2 has a number

**=SUMPRODUCT((ISTEXT(Range1)\*(ISNUMBER(Rang2))))**

*Example: Use this function to count cells only if Col.1 is text and Col.2 is number*

Col.1	Col.2
a	1
b	2
c	3
	5
e	3
f	

Ref:140509

4
---

Formula Used: =SUMPRODUCT((ISTEXT(C98:C104)\*(ISNUMBER(E98:E104))))

### COUNTIFS - Count rows that meet two or more criteria.

**=COUNTIFS(J26:J31,J33,K26:K31,">=" &K33)**

Col.1	Col.2
Pen	1
Pencil	2
Binder	3
Pen	20
Pen	15
Pencil	

temp

Ref: 140508

Pen	10
-----	----

Count includes Pen, Gel Pen, Pencil

2
---

=COUNTIF(F52:F56, "\*Pen\*")

## SUMPRODUCT Function

**=SUMPRODUCT(array1, array2,...)**

**Example 1**

	Value A	Value B
Item 1	2	1
Item 2	3	1
Item 3	4	2

13

Formula Used: =SUMPRODUCT(E10:E12,F10:F12)  
=(E10xF10)+(E11xF11)+(E12xF12)

**Example 2**

	Value A	Value B
Item 1	2	2
Item 2	3	1
Item 3	4	2

14

Formula Used: =SUMPRODUCT(E19:E21+F19:F21)  
=(E19+F19)+(E20+F20)+(E21+F21)

**Example 3**

	Value A	Value B
Item 1	2	2
Item 2	3	1
Item 3	4	2

6

Formula Used: =SUMPRODUCT(E28:E30/F28:F30)  
=(E28/F28)+(E29/F29)+(E30/F30)

**Example 4 - Sumproduct if meet Criteria**

	Value A	Value B
Apple	2	2
Orange	3	1
Apple	4	2

12

Formula Used: =SUMPRODUCT((E37:E39\*F37:F39)\*(D37:D39="Apple"))  
=(E37\*F37)\*True+(E38\*F38)\*False+(E39\*F39)\*True

**Example 5 - Sumproduct if meet Multiple Criteria**

		Value A	Value B
A	Apple	2	2
B	Orange	3	1
A	Apple	4	2
B	Apple	4	2

12

Formula Used: =SUMPRODUCT((E46:E49\*F46:F49)\*(D46:D49="Apple")\*(C46:C49="A"))



### ROUND Function

**=ROUND(number, digits)**

Use this function to round number to various places.

*Examples:*

	<b>12,345.6789</b>	
<i>Round to three decimal places</i>	12,345.6790	Formula Used: =ROUND(D9,3)
<i>Round to two decimal places</i>	12,345.6800	Formula Used: =ROUND(D9,2)
<i>Round to one decimal place</i>	12,345.7000	Formula Used: =ROUND(D9,1)
<i>Round to ones place</i>	12,346.00	Formula Used: =ROUND(D9,0)
<i>Round to nearest multiple of 10</i>	12,350.00	Formula Used: =ROUND(D9,-1)
<i>Round to nearest multiple of 100</i>	12,300.00	Formula Used: =ROUND(D9,-2)
<i>Round to nearest multiple of 1000</i>	12,000.00	Formula Used: =ROUND(D9,-3)

### ROUNDUP Function

**=ROUNDUP(number, digits)**

Use this function to round number up to various places.

*Examples:*

	<b>1,234.6710</b>	
<i>Round to three decimal places</i>	1,234.6710	Formula Used: =ROUNDUP(D25,3)
<i>Round to two decimal places</i>	1,234.6800	Formula Used: =ROUNDUP(D25,2)
<i>Round to one decimal place</i>	1,234.7000	Formula Used: =ROUNDUP(D25,1)
<i>Round to ones place</i>	1,235.00	Formula Used: =ROUNDUP(D25,0)
<i>Round to nearest multiple of 10</i>	1,240.00	Formula Used: =ROUNDUP(D25,-1)
<i>Round to nearest multiple of 100</i>	1,300.00	Formula Used: =ROUNDUP(D25,-2)
<i>Round to nearest multiple of 1000</i>	2,000.00	Formula Used: =ROUNDUP(D25,-3)

### ROUNDDOWN Function

**=ROUNDDOWN(number, digits)**

Use this function to round number down to various places.

*Examples:*

	<b>1,235.6790</b>	
<i>Round to three decimal places</i>	1,235.6790	Formula Used: =ROUNDDOWN(D41,3)
<i>Round to two decimal places</i>	1,235.6700	Formula Used: =ROUNDDOWN(D41,2)
<i>Round to one decimal place</i>	1,235.6000	Formula Used: =ROUNDDOWN(D41,1)
<i>Round to ones place</i>	1,235.00	Formula Used: =ROUNDDOWN(D41,0)
<i>Round to nearest multiple of 10</i>	1,230.00	Formula Used: =ROUNDDOWN(D41,-1)
<i>Round to nearest multiple of 100</i>	1,200.00	Formula Used: =ROUNDDOWN(D41,-2)
<i>Round to nearest multiple of 1000</i>	1,000.00	Formula Used: =ROUNDDOWN(D41,-3)

### MROUND Function

**=MROUND(number, multiple)**

Use this function to round number to the desired multiple.

*Example 1*

	<b>12,345</b>	
<i>Rounds up to nearest multiple of 2</i>	12,346	Formula Used: =MROUND(D55,2)
<i>Rounds up to nearest multiple of 5</i>	12,345	Formula Used: =MROUND(D55,5)
<i>Rounds up to nearest multiple of 10</i>	12,350	Formula Used: =MROUND(D55,10)
<i>Rounds up to nearest multiple of 100</i>	12,300	Formula Used: =MROUND(D55,100)

## TEXT Functions

*Example 1:*

hello, this is your INSTRUCTOR.

**=LOWER(text)**

hello, this is your instructor.

Formula Used: =LOWER(C6)

**=UPPER(text)**

HELLO, THIS IS YOUR INSTRUCTOR.

Formula Used: =UPPER(C6)

**=PROPER(text)**

Hello, This Is Your Instructor.

Formula Used: =PROPER(C6)

*Example 2:*

Bananna

**=LEFT(text, number)**

Ban

Formula Used: =LEFT(C19,3)

**=RIGHT(text, number)**

nna

Formula Used: =RIGHT(C19,3)

**=MID(text, number)**

nan

Formula Used: =MID(C19,3,3)

**=REPLACE(xxx)**

BXXanna

Formula Used: =REPLACE(C19,2,2,"XX")

**=LEN(text)**

7

Formula Used: =LEN(C19)

Find Date of Week		
4/15/2015	Wednesday	Formula Used: =TEXT(B5,"dddd")
8/25/2016	Thu	Formula Used: =TEXT(B6,"ddd")

Find Month		
7/30/2016	7	Formula Used: =MONTH(B9)

Find Week Number		
8/21/2016	35	Formula Used: =WEEKNUM(B12)

Find the Sunday of		
4/15/2015	4/12/2015	Formula Used: =B15-WEEKDAY(B15)+1

Find Number of Days, Months, & Years Between Two Dates		
Start:	1/12/2015	
End:	7/30/2016	
Days =	565	Formula Used: =DATEDIF(C18,C19,"d")
Months =	18	Formula Used: =DATEDIF(C18,C19,"m")
Years =	1	Formula Used: =DATEDIF(C18,C19,"y")

Find Number of Workdays Between Two Dates		
Start:	1/1/2012	
End:	5/5/2015	
	872	Formula Used: =NETWORKDAYS(C25,C26)

Find Number of Weeks Between Two Dates		
Start:	1/1/2012	
End:	9/25/2015	
	194	Formula Used: =INT((C31-C30)/7)

Check If Dates Are in The Same Month & Year		
Start:	9/2/2015	
End:	9/25/2015	
	TRUE	Formula Used: =MONTH(C35)&YEAR(C35)=MONTH(C36)&YEAR(C36)

Find Number of years, months and days Between Two Dates		
Start:	5/21/1978	
End:	11/26/2013	
	35 year(s), 6 month(s) and 5 day(s)	Formula Used: =IF(C41,IF(AND(DATEDIF(C40,C41,"y") <=0, DATEDIF(C40,C41,"ym") <=0), DATEDIF(C40,C41,"md") & " day(s)", IF(DATEDIF(C40,C41,"y")<=0, DATEDIF(C40,C41,"ym") & " month(s) and "& DATEDIF(C40,C41,"md") & " day(s)", DATEDIF(C40,C41,"y") & " year(s), " & DATEDIF(C40,C41,"ym") & " month(s) and "& DATEDIF(C40,C41,"md") & " day(s)")), "")

Display Month		
7/30/2016	7	Formula Used: =MONTH(B9)

4/6/2015	Today is 04/06/2015	Formula Used: ="Today is " & TEXT(B48, "MM/DD/YYYY")
----------	---------------------	--

**Text and Numbers in One Cell**

**= "TEXT 1" & " TEXT(Number)**

Use this function combine text and numbers into one cell

*Examples*

12/31/2013	Payment is due 12/31/2013	Formula Used: ="Payment is due " & TEXT(C9,"mm/dd/yyyy")
150	Amount due: \$150.00 USD	Formula Used: ="Amount due: " & TEXT(C10,"\$#,##0.00")& " USD"
2,678,678	Total: 2,678,678.00	Formula Used: ="Total: "&TEXT(C11,"#,##0.00")
0.8	Your score is 80.00%	Formula Used: ="Your score is " & TEXT(C12,"0.00%")
0.85	Your score is 85%	Formula Used: ="Your score is " & TEXT(C13,"0%")
4.75	Hours worked: 4 3/4	Formula Used: ="Hours worked: " &TEXT(C14,"# ?/?")

1256.3	Total: \$1,256.30	Formula Used: ="Total: "&TEXT(C16,"\$#,##0.00")
--------	-------------------	---

Report printed on February 4, 2018 at 8:41 PM	Formula Used: ="Report printed on "&TEXT(NOW(),"mmmm d, yyyy at h:mm AM/PM")
Today is Sunday	Formula Used: ="Today is " & TEXT(TODAY(),"dddd")

**ISERR Function****=IF(ISERR(ref),0,ref)**

Use this function to replace #REF!, #DIV/0!, etc with 0

**Example 1: Sum cells in continuous column**

#DIV/0!
0

Formula Used: =IF(ISERR(ref),0,ref)

**Example 2: Sum cells in non-continuous column or row**

3	
6	
	6
6	
21	

Formula Used: =SUM(C16:C17,D18,C19)

**SUM Function**

**=SUM(Range)**  
**=SUM(Range, Cell or Range)**

Use this function to sum cells.

*Example 1: Sum cells in continuous column*

5
5
5
15

Formula Used: =SUM(C10:C12)

*Example 2: Sum cells in non-continuous column or row*

3	
6	
	6
6	
21	

Formula Used: =SUM(C16:C17,D18,C19)

**SUMIF Function**

**=SUMIF(Range, criteria, [sum\_range])**

Use this function to sum cells based on criteria.

*Example 1: Sum cells with values only (ignore text and #value!)*

5
9
5
19

Formula Used: =SUMIF(C30:C32, ">0")

*Example 2: Sum Column 1 if Column 2 matches criteria*

Column 1	Column 2
2	Oranges
5	Apples
4	Apples
6	Oranges
2	Oranges
5	Apples

Oranges:	10
Apples:	14

Formula Used: =SUMIF(D38:D43,"Oranges",C38:C43)

Formula Used: =SUMIF(D38:D43,"Apples",C38:C43)

## IF Function

**=IF(Condition1, Value\_If\_True, Value\_If\_False)**

Use this function to evaluate a condition and output value depending on true or false of condition.

*Example 1: True only if cell equals to 5 and output "Good". Else output "Bad"*

5
Good

Formula Used: =IF(C9=5, "Good", "Bad")

*Example 2: True only if cell equals to 5 and output Value\_If\_True. Else output Value\_If\_False*

5
4

Formula Used: =IF(C13>5, C13+1, C13-1)

## NESTED IFs Function

**=IF(Condition1, Value\_If\_True1, IF(Condition2, Value\_If\_True2, Value\_If\_False2))**

Use this function to evaluate two or more conditions and output value depending on true or false of conditions.

*Example 1: If Condition1 is True, output "Good".*

*If Condition1 is False, evaluate Condition2 and output Value\_If\_True2 or Value\_If\_False2.*

1
2
Good

Formula Used: =IF(C24=1, "Good", IF(C25=2, "Better", "Best"))

**AND Function****=AND(condition1, condition2, condition3, etc.)**

Use this function to output TRUE only if ALL of the conditions were true.

*Example 1: True if all cell values equal to 5*

5
5
5
TRUE

Formula Used: =AND(C9=5, C10=5, C11=5)

*Example 2: True if cells C15<5, C16>5, and C17>5*

3
6
6
TRUE

Formula Used: =AND(C15&lt;5, C16&gt;5, C17&gt;5)

**OR Function****=OR(condition1, condition2, condition3, etc.)**

Use this function to output TRUE if any one of the condition is true.

*Example 1: True if any cell equal to 5*

1
5
1
TRUE

Formula Used: =OR(C27=5, C28=5, C29=5)



SUM EVERY OTHER ROWS

**=SUMPRODUCT((MOD(ROW(range),2)=0)\*(range))**

Use this function to sum every other row (all odd or even).

**Example 1**

A	1
B	2
A	1
B	2
A	1
B	2
A	1
B	2
A	1
B	2
A	1
B	2
A	1
B	2
A	1
B	2

Sum all Odd Rows (A)	8
Sum all Even Rows (B)	16

Formula Used: =SUMPRODUCT((MOD(ROW(D9:D24),2)=1)\*(D9:D24))

Formula Used: =SUMPRODUCT((MOD(ROW(D9:D24),2)=0)\*(D9:D24))

[Ref: 140504]

## INDEX Function

**=INDEX(array, row number, column number)**

Use this function to output cell value in specific Row # and Column #.

### Example 1

	Col. 1	Col. 2	Col. 3	Col. 4	Col. 5	Col. 6
Row 1	10	20	30	40	50	60
Row 2	11	21	31	41	51	61
Row 3	12	22	32	42	52	62
Row 4	13	23	33	43	53	63
Row 5	14	24	34	44	54	64
Row 6	15	25	35	45	55	65
Row 7	16	26	36	46	56	66
Row 8	17	27	37	47	57	67
Row 9	18	28	38	48	58	68

Row #	Column #	Result
8	5	57

Formula Used: =INDEX(D10:I18, D21,E21)

## 2D Array Search Function

**=INDEX(array, row number, column number)**

Use Index function with Match functions to output cell value in Row and Column with matching criteria.

*Example 1: INDEX(array, MATCH(K25,C23:C31,), MATCH(L25,D22:I22,))*

	A	B	C	D	E	F
1	10	20	30	40	50	60
2	11	21	31	41	51	61
3	12	22	32	42	52	62
4	13	23	33	43	53	63
5	14	24	34	44	54	64
6	15	25	35	45	55	65
7	16	26	36	46	56	66
8	17	27	37	47	57	67
9	18	28	38	48	58	68

Row #	Column #	Result
3	E	52

Formula Used: =INDEX(D11:I19, MATCH(D22,C11:C19,), MATCH(E22,D10:I10,))

## Strings In Cell

*Examples*

	First Second Third and Last	
Return first letter in a string	F	Formula Used: =LEFT(C8, FIND(" ", C8, 1))
Return first word in a string	First	Formula Used: =LEFT(C8, FIND(" ", C8, 1))
Return second word in a string	Second	Formula Used: =MID(C8, FIND(" ", C8, 1)+1, FIND(" ", C8, FIND(" ", C8, 1)+1)-(FIND(" ", C8, FIND(" ", C8, 1))))
Return all but the first word in a string	Second Third and Last	Formula Used: =RIGHT(C8, LEN(C8)-FIND(" ", C8, 1))

## Text and Numbers In Same Cell

**Examples**

12/31/2013	Payment is due 12/31/2013
150	Amount due: \$150.00 USD
0.8	Your score is 80.00%
4.75	Hours worked: 4 3/4

Formula Used: ="Payment is due " & TEXT(B8,"mm/dd/yyyy")

Formula Used: ="Amount due: " & TEXT(B9,"\$#,##0.00")& " USD"

Formula Used: ="Your score is " & TEXT(B10,"0.00%")

Formula Used: ="Hours worked: " &TEXT(B11,"# ?/?")

1256.3	Total: \$1,256.30
--------	-------------------

Formula Used: ="Total: "&TEXT(B34,"\$#,##0.00")

Report printed on February 4, 2018 at 8:41 PM
Today is Sunday

Fomula Used: ="Report printed on "&TEXT(NOW(),"mmm d, yyyy at h:mm AM/PM")

Formula Used: ="Today is " & TEXT(TODAY(),"dddd")

## Basic Formulas

MAIN

FORMULA	DESCRIPTION
=	Equal To
<>	Not Equal To
>	Greater Than
<	Less Than
>=	Greater Than or Equal to
<=	Less Than or Equal to
/	Divide
*	Multiply
=MIN(A1:A100)	Output the smallest number in a range (Cells A1 to A100)
=MAX(A1:A100)	Output the largest number in a range (Cells A1 to A100)
=Small(A1:A100,2)	Output the second smallest number in a range (Cells A1 to A100)
=Large(A1:A100,5)	Output the fifth largest number in a range (Cells A1 to A100)
=SUM(A1:A100)	Add cells A1 to A100
=SUM(A1, A5, A7)	Add cells A1, A5 and A7
=SUM(A1+A5+A7)	Add cells A1, A5 and A8
=SUMIF(A1:A10; ">10")	Sum all number in range A1 to A10 if the numbers are greater than 10.
=COUNTIF(A1:A10; ">10")	Count all number in range A1 to A10 if the numbers are greater than 10.
=COUNT(A1:A10)	Count the numbers in a range, ignores blank or empty cells.
=COUNTA(A1:A10)	Count all character in a range, ignores blank or empty cells.
=TODAY()	Output today's date
=AVERAGE(A1:A10)	Output the average of the range, cells A1 to A10
=EXACT(text1, text2)	Return TRUE if both are same
=CONCATENATE(A1,A5,A7)	Join strings together. Join strings in cell A1, A5 and A7.
=UPPER(A1)	Convert the text in cell A1 to all upper case letters.
=LOWER(A1)	Convert the text in cell A1 to all lower case letters.
=AND(A1=1, B1=1,1,0)	AND
=OR(A1=1, B1=1,1,0)	OR
=IF(AND(A1=1,B1=1),1,0)	Nested If AND
=IF(OR(A1=1,B1=1),1,0)	Nested If OR

## LOOKUP Function

**=LOOKUP(2,1/((criteria1)\*(criteria2)\*(criteria3)),array)**

*Example 1: Output cell value if 3 criteria were met*

Match 1	Match 2	Match 3	Output
a	1	0	200
b	2	1	210
c	3	2	220
d	4	3	230
e	5	4	240
f	6	5	250
g	7	6	260
h	8	3	270
i	9	8	280
j	10	9	290
k	11	10	300
l	12	11	310
m	13	12	320
n	14	13	330

**230**

Formula Used: =LOOKUP(2,1/((C9:C23="d")\*(D9:D23=4)\*(E9:E23=3)),F9:F23)

*Example 2*

First	Last	Zip	Income
John	Smith	11356	\$52,000
Victoria	Xu	11326	\$26,000
Valerie	Lee	10001	\$65,000
John	Lee	12542	\$105,000
Sunny	Wong	10055	\$36,500
Jay	Wu	10033	\$12,562
John	Lee	10025	\$45,200
Jay	Chan	10065	\$36,950
Amy	Chen	10058	\$89,000
Jack	Li	10065	\$79,000
Sandy	Smith	15520	\$36,500
Sandy	Smith	12566	\$25,690
Jim	Lau	14558	\$50,000
Tim	Lai	16559	\$250,000

First	Sandy
Last	Smith
Zip	12566

**\$25,690**

Formula Used: =LOOKUP(2,1/((C28:C42=D44)\*(D28:D42=D45)\*(E28:E42=D46)),F28:F42)

## Shortcut Keys

MAIN

Key	Description
<b>CTRL+S</b>	Save
<b>CTRL+Z</b>	Undo
<b>CTRL+;</b>	Insert today's date
<b>CTRL+C</b>	Copy
<b>CTRL+V</b>	Paste
<b>CTRL+P</b>	Print
<b>CTRL+A</b>	Select all (when you are not entering or editing a formula)
<b>CTRL+SPACEBAR</b>	Select the current column
<b>SHIFT+SPACEBAR</b>	Select the current row
<b>CTRL+ '</b>	Alternate between displaying cell values and displaying cell formulas
<b>F9</b>	Calculate all sheets in all open workbooks
<b>SHIFT+F9</b>	Calculate the active worksheet
<b>F11 or ALT+F1</b>	Create a chart that uses the current range
<b>CTRL+1</b>	Display the Format Cells dialog box
<b>F5</b>	Display the Go To dialog box
<b>CTRL+ENTER</b>	Fill the selected cell range with the current entry
<b>CTRL+:</b>	Insert the current time
<b>CTRL+HOME</b>	Move to the beginning of the worksheet
<b>CTRL+END</b>	Move to the last cell on the worksheet, which is the cell at the intersection of the rightmost used column and the bottommost used row (in the lower-right corner), or the cell opposite the home cell, which is typically A1
<b>CTRL+O</b>	Open
<b>SHIFT+F3</b>	Paste a function into a formula
<b>CTRL+A</b>	When you enter a formula, display the Formula Palette after you type a function name
<b>CTRL+Mouse Scroll</b>	Zoom In or Zoom Out



V-Lookup Example

Cost Calculation Sheet				
	Item	Quantity	Unit Price	Line Total
1	Apple	10	\$1.00	\$10.00
2	Bananna	15	\$2.00	\$30.00
3	Apple	30	\$1.00	\$30.00
4	Bananna	10	\$2.00	\$20.00
5	Apple	5	\$1.00	\$5.00
6	Apple	56	\$1.00	\$56.00
7	Apple	2	\$1.00	\$2.00
8	Apple	3	\$1.00	\$3.00
<b>Grand Total =</b>				<b>\$156.00</b>

Unit Pricing Table	
Item	Unit Price
Apple	\$1.00
Orange	\$1.50
Bananna	\$2.00
Pear	\$3.00
Watermelon	\$7.00
Kiwi	\$1.10
Grapes	\$2.00
Strawberry	\$3.00

**Example:**  
 Above is a quick practical example using vlookup function with dropdown list to generate unit price

V-LookUp Function

=VLOOKUP(value, table, index\_number, [not\_exact\_match])

[not\_exact\_match] Optional - Enter FALSE to find exact match. Enter TRUE to find an approximate match (next largest value that is less than value.)

Example 1

ID Number	Item	Qty	Price	Total
1001	Apple	10	\$1.00	\$10.00
1002	Orange	5	\$2.00	\$10.00
1003	Pear	25	\$3.00	\$75.00
1004	Banana	50	\$4.00	\$200.00
1006	Pineapple	15	\$5.00	\$75.00
1007	Peach	14	\$6.00	\$84.00
1008	Strawberry	50	\$7.00	\$350.00

1004	Banana
------	--------

Formula Used: =VLOOKUP(I11,C11:E17,2,FALSE)

1005	Banana
------	--------

Formula Used: =VLOOKUP(I13,C11:E17,2)

Example 2

ID	Name	A	B	C
101	Amy	1.1	1.2	1.3
102	Ben	2.1	2.2	2.3
103	Cindy	3.1	3.2	3.3
104	Dave	4.1	4.2	4.3
105	Eric	5.1	5.2	5.3
106	Frank	6.1	6.2	6.3
120	vicky	7.1	6.2	6.3
108	Henry	8.1	8.2	8.3
109	Ivan	9.1	9.2	9.3
110	ERROR			

ID	Name	A	B	C
102	Ben	2.1	2.2	2.3

Notes:

ID column data must be in ascending order

No match, use previous line data

Example 3

Col. 1	Col. 2	Col. 3	Col. 4	Col. 5	Col. 6	Col. 7
A	1.1	2.1	3.1	4.1	5.1	6.1
B	1.2	2.2	3.2	4.2	5.2	6.2
C	1.3	2.3	3.3	4.3	5.3	6.3
D	1.4	2.4	3.4	4.4	5.4	6.4
E	1.5	2.5	3.5	4.5	5.5	6.5
F	1.6	2.6	3.6	4.6	5.6	6.6
G	1.7	2.7	3.7	4.7	5.7	6.7
H	1.8	2.8	3.8	4.8	5.8	6.8
I	1.9	2.9	3.9	4.9	5.9	6.9

Row	Column	Result
C	6	5.3

Notes:

If No Match, uses previous line data

## MATCH Function

**=MATCH(Value, Array, Match\_type)**

**Match\_type**

1 (default) - Find largest value that is less than or equal to value. Data needs to be sorted in ascending order.

0 - Find the first value that is equal to value. Data can be sorted in any order.

-1 - Find the smallest value that is greater than or equal to value. Data needs to be sorted in descending order.

**Match\_type = 1 (Default)**

	Array
Position 1	10
Position 2	11
Position 3	12
Position 4	13
Position 5	14
Position 6	15
Position 7	16
Position 8	17
Position 9	18

Array Value	Position Result
17	8

Formula Used: =MATCH(C25,D14:D22,1)

**Match\_type = 0**

	Array
Position 1	10
Position 2	11
Position 3	17
Position 4	13
Position 5	14
Position 6	15
Position 7	16
Position 8	17
Position 9	18

Array Value	Position Result
17	3

Formula Used: =MATCH(C40,D29:D37,0)

**Match\_type = -1**

	Array
Position 1	20
Position 2	19
Position 3	18
Position 4	17
Position 5	16
Position 6	15
Position 7	14
Position 8	13
Position 9	12

Array Value	Position Result
17	4

Formula Used: =MATCH(C55,D44:D52,0-1)